The Hairer − Wanner DAE

Simplifying a polynomial DAE (an implicit ODE ?) to get the variety of the constraints and the underlying ODE

```
> restart;
  with (DifferentialAlgebra0):
> R := DifferentialRing (derivations = [t], blocks = [[y, x, s, c, z], [a, b]], parameters = [a,b],
  notation = diff);
```

$$R := differential\_ring \quad (1)$$

Initially, a steering wheel story featuring x(t), y(t) and z(t)
s(t) encodes sin (b*z(t))
c(t) encodes cos (b*z(t))

```
> syst := [ x[t] = a*y + s, y[t] = 2*a*x + c, x^2 + y^2 = 1,
           s[t] = b*z[t]*c, s^2 + c^2 = 1 ];
```

$$syst := \left[x_t = a\,y + s,\, y_t = 2\,a\,x + c,\, x^2 + y^2 = 1,\, s_t = b\,z_t\,c,\, s^2 + c^2 = 1\right] \quad (2)$$

The same system, in usual notation

```
> syst := NormalForm (syst, R, input = jet);
```

$$syst := \left[\frac{d}{dt}\,x(t) - y(t)\,a - s(t),\, \frac{d}{dt}\,y(t) - 2\,x(t)\,a - c(t),\, y(t)^2 + x(t)^2 - 1,\, \frac{d}{dt}\,s(t) - \left(\frac{d}{dt}\,z(t)\right)c(t)\,b,\, s(t)^2 + c(t)^2 - 1\right] \quad (3)$$

One tunes the regular differential chain (= characteristic set) we are looking for.
One computes the "general" solution only

```
> ideal := RosenfeldGroebner (syst, R, basefield = field (generators = [a, b]), attributes =
  [differential, primitive], singsol=none);
```

$$ideal := [regular\_differential\_chain] \quad (4)$$

```
> Equations (ideal[1]);
```

$$\left[s(t)^2 + c(t)^2 - 1,\, 9\,x(t)^4\,a^2 + 6\,x(t)^3\,c(t)\,a - 9\,x(t)^2\,a^2 + x(t)^2 - 6\,x(t)\,c(t)\,a - c(t)^2,\, 3\,y(t)\,x(t)\,a + y(t)\,c(t) + x(t)\,s(t),\, 3\left(\frac{d}{dt}\,z(t)\right)x(t)^3\,c(t)\,a\,b + \left(\frac{d}{dt}\,z(t)\right)x(t)^2\,b + 12\,x(t)^3\,c(t)\,a^2 + 9\,x(t)^2\,c(t)^2\,a + 18\,x(t)^2\,a^3 - 2\,x(t)^2\,a + 9\,x(t)\,c(t)\,a^2 + x(t)\,c(t) + c(t)^2\,a,\, \frac{d}{dt}\,c(t) + \left(\frac{d}{dt}\,z(t)\right)s(t)\,b\right] \quad (5)$$

The equations of order 0 give the variety of the constraints

```
> Equations (ideal[1], order=0);
```

$$\left[s(t)^2 + c(t)^2 - 1,\, 9\,x(t)^4\,a^2 + 6\,x(t)^3\,c(t)\,a - 9\,x(t)^2\,a^2 + x(t)^2 - 6\,x(t)\,c(t)\,a - c(t)^2,\, 3\,y(t)\,x(t)\,a + y(t)\,c(t) + x(t)\,s(t)\right] \quad (6)$$

Here is the missing equation, in order to build the underlying ODE

```
> Equations (ideal[1], solved, leader=diff (z(t),t));
```

$$\left[\frac{d}{dt}\,z(t) = -\frac{1}{3\,x(t)^3\,c(t)\,a\,b + x(t)^2\,b}\left(12\,x(t)^3\,c(t)\,a^2 + 9\,x(t)^2\,c(t)^2\,a + 18\,x(t)^2\,a^3\right. \quad (7)$$

$$- 2\,x(t)^2\,a + 9\,x(t)\,c(t)\,a^2 + x(t)\,c(t) + c(t)^2\,a)$$